Appendix A
Appendix A.1. SAS Code Used in the Original Analysis
Data import and preprocessing
filename myfile '/home/u64227463/Project/HOST_Dataset_May_2025.xlsx';
proc import datafile=myfile out=host_data dbms=xlsx replace;
    sheet='SPSS-1 Full';
    getnames=yes;
run;

data host_analyzed;
    set host_data;
    log_JADAS = log(JADAS + 1);
run;

Descriptive statistics
proc means data=host_analyzed n mean std median min max nmiss;
    var JADAS ESR CRP HPTG CalcAlpha2 AgeTest;
run;

Pearson correlation analysis
proc corr data=host_analyzed pearson nosimple;
    var ESR CRP HPTG CalcAlpha2 AgeTest;
    with JADAS;
run;

Diagnostic plots
proc glm data=host_analyzed plots=diagnostics;
    model JADAS = ESR CRP HPTG CalcAlpha2 AgeTest;
run;

proc glm data=host_analyzed plots=diagnostics;
    model log_JADAS = ESR CRP HPTG CalcAlpha2 AgeTest;
run;

Stepwise regression and final model
proc reg data=host_analyzed;
    model log_JADAS = ESR CRP HPTG CalcAlpha2 AgeTest / selection=stepwise slentry=0.05;
run;
quit;

proc reg data=host_analyzed;
    model log_JADAS = ESR HPTG / ss2;
run;
quit;

Gender interaction and confounding checks
proc glm data=host_analyzed;
    class GenderNum;
    model log_JADAS = ESR HPTG GenderNum ESR*GenderNum / ss3;
run;

proc glm data=host_analyzed;
    class GenderNum;
    model log_JADAS = ESR HPTG GenderNum HPTG*GenderNum / ss3;
run;

proc reg data=host_analyzed;
    model log_JADAS = ESR HPTG;
run;

proc reg data=host_analyzed;
    model log_JADAS = ESR HPTG GenderNum;
run;
quit;


